

מבצע BUG
זיכרון ב:
APPLE

ההשבת



12345

משהו על חברת "מחשבת" – מערכות למידה

זו חברה, אשר אחד מתחומי התמחותה העיקריים הוא פיתוח מערכות לימוד עצמי.

מחברי יחידות לימוד אלה עוסקים במשך שנים בפיתוח מערכות ללימוד עצמי עבור המגזר האזרחי והצבאי.

חל איסור מוחלט להעתיק או לשכפל יחידת לימוד זו בשלמותה או בחלקה לכל מטרה שהיא או לעשות בה שימוש מסחרי כלשהו, ללא רשות בכתב מאת חברת מחשבת – מ.ל. בע"מ.

© 1986 – כל הזכויות שמורות למחשבת – מ.ל. בע"מ.

ת.ד. 48032 ת"א מיקוד 61480.

הוצאה לאור – מחשבת מ.ל. בע"מ.

דפוס: דפוס וכריכת מאיר

יהודה הלוי 4, חל'אביב, טלפון: 03-650455

Copyright © 1986 by MAKH-SHEVET M.L. Ltd.

P.O.Box 48032 Tel-Aviv 61480, Israel

APPLE
II-C, II-E, II, II+

BUG

MICROCOMPUTERS BOOKS & SOFTWARE

המרכז לספרות עזר ותוכנה למחשבים

מחשבת

צעדים מתקדמים בתכנות
בלימוד עצמי

יחידה 5

גרפיקה מתמטית

תוכן העניינים

עמוד

5 פתח דבר
6 פרק א - סינוסים וקוסינוסים
22 פרק ב - שיעון מחוגים
24 פרק ג - סינוסואידות
29 פרק ד - ציורי ליסאז'ו
32 פרק ה - מצולעים משוכללים
39 פרק ו - צמצמים
44 נספח א - זוויות ומשולשים
 נספח ב - סינוסים וקוסינוסים של זוויות הגדולות
48 מ-90 מעלות
52 נספח ג - הסבר ציורי סינוסואידות
54 נספח ד - תוכנית הדגמה לציורי ליסאז'ו
58 תשובות
70 רשימת מושגים נלמדים

פתח דבר

חוברת זאת הנקראת "גרפיקה מתמטית" מוקדשת לפיתוח תוכניות גרפיות מרהיבות.

כדי לעשות זאת איך אפשרות להתחמק משימוש בכמה כלים מתמטיים מתחום הטריגונומטריה: הסינוס והקוסינוס.

אל דאגה! אינך צריך להכיר מושגים אלה לפני שאתה מתחיל ללמוד. אנחנו נלמד אותך צעד אחר צעד כל מושג מתמטי חדש שאנו עושים בו שימוש.

כשתגיע לסיומה של יחידת לימוד זו לא רק שתדע לפתח תוכניות מחשב, אלא גם תעשיר את הרקע המתמטי שלך.

טעות לעולם חוזרת...

אם יש לך בעיה הקשורה בחומר, או שמצאת טעות, נשמח אם תתקשר ישירות אלינו בטלפון: 03-492723.

בברכת המשך לימוד

נעים ומרתק

דני קדם ויצחק קליסקי

"מחשבת" - מערכות למידה

פרק א

סינוסים וקוסינוסים


מבוא

נפתח את הפרק בתוכנית גרפית אשר בה יופיעו שלושה "יצורים"


חדשים: SIN - סינוס

COS - קוסינוס

והמספר: 3.14159265

העתק את התוכנית הבאה למחשב והרץ אותה: 

```
5 HGR : HCOLOR = 3
10 PI = 3.14159265
20 FOR M=0 TO 360 STEP 5
30 z=PI/180*M
40 a=60*COS(z)
50 b=60*SIN(z)
60 HPLOT 140+a, 80+b
70 NEXT m
```

מה שרטט המחשב על המסך? 

קרוב לודאי שהמחשב רשם על המסך נקודות שיצרו ____ (ריבוע, מעגל).
בפרק זה נרד לעומקם של המושגים החדשים שנכנסו לתוכנית ונראה
כי הם יאפשרו לך לפתח תוכניות גרפיות עשירות ויפות ביותר.

שאלת רקע

לפניך משולש ____ (ישר זווית,

קהה זווית).


הצלעות ____ ו- ____ הם הניצבים

ואילו צלע ____ היא היתר.

הניצב מול הזווית (m) הוא הצלע ____.


אם אינך יודע לענות על שאלה זאת פנה ללמוד את נספח א,

בעמוד 44.



סינוס (SIN)

בשלושת המשולשים ישרי הזווית שבאיור הבא נמצאים הניצבים b_1 , b_2 ו- b_3 מול זווית של 30° מעלות:
(שים לב כי כאשר כתוב 30° הכוונה ל- 30 מעלות)



☐ קח סרגל ומדוד על השרטוט את אורך הניצב b_1 והיתר c_1 במשולש

ישר הזווית הקטן: $c_1 = \underline{\quad ? \quad}$ $b_1 = \underline{\quad ? \quad}$

☐ מדוד את האורכים במילימטרים ונסה לדייק ככל האפשר במדידותיך.

פקוד על המחשב לחשב ולהדפיס את הערך של:

$$\frac{b_1}{c_1} = ?$$

(ביחידה 1 פרק ב' עסקת בביצוע תרגילים כאלה)

☐ חזור ומדוד בעזרת סרגל את האורכים של הניצב והיתר בשני

המשולשים הנותרים: $c_2 = \underline{\quad}$ $b_2 = \underline{\quad}$

$c_3 = \underline{\quad}$ $b_3 = \underline{\quad}$

ופקוד על המחשב להדפיס על המסך את הערכים של $\frac{b_2}{c_2} = ?$


ו- $\frac{b_3}{c_3} = ?$

התכונן בשלושת המספרים הרשומים עכשיו על המסך.
אם דייקת במדידותיך קרוב לודאי ששלושת המספרים כמעט שווים
האחד לשני.

המספרים על המסך שווים או קרובים מאוד ל- (0.4, 0.5, 0.6) ?
(0.7)

(תשובה 1)

לפניך משולש ישר זווית שאחת מזוויותיו היא 30° ?



בלי למדוד - האם תוכל לדעת למה
שווה היחס בין אורך הניצב b , שמול
הזווית 30° , לבין אורך היתר c ?

$$\frac{b}{c} = \underline{\hspace{2cm}}$$

אנו מתארים לעצמנו כי קבעת, גם ללא מדידה, שהיחס $\frac{b}{c}$ שווה
ל-0.5. הסיבה לכך היא שמשולש זה דומה לכל אחד מהמשולשים
הקודמים ישרי הזווית בעלי זווית 30° שעסקנו בהם. ולכן גם
לגביו מתקיים אותו יחס שמצאת בשלושת המשולשים הקודמים.

נסכם

בכל משולש ישר זווית, אשר אחת הזוויות החדות שלו היא 30°
היחס בין אורך הניצב הנמצא מול זווית זאת לבין אורך היתר
יהיה תמיד 0.5 - לא חשוב מה גודלו של המשולש! (אם תוצאות
המדידה שלך היו קצת שונות - הסיבה נובעת מאי-דיוקים קלים
במדידה עם הסרגל.)

ליחס שבין אורך הניצב שמול הזווית לבין אורך היתר קוראים בשם:
סינוס של הזווית (SIN)


במקרה שלנו אפשר להגיד כי: "סינוס של 30° שווה ל-0.5".
וכותבים זאת כך:

$$\text{SIN}(30^\circ) = \frac{\text{אורך הניצב מול } 30^\circ}{\text{אורך היתר}} = \frac{1}{2} = 0.5$$

האם תוכל באותו אופן שחישבת את $\text{SIN}(30^\circ)$ לחשב, בעזרת
המשולשים המשורטטים בעמוד הקודם, את הערך של סינוס 60° ?

$$\text{SIN}(60^\circ) = \frac{\text{אורך הניצב מול הזווית } 60^\circ}{\text{אורך היתר}} = \underline{\hspace{2cm}}$$

רמז: מתוך נספח 1 היית צריך לדעת כי בכל משולש סכום הזוויות שווה ל- .
עדיין מתקשה? פנה לתשובה 2.



נסכם: ניתן לומר באופן כללי, כי סינוס הזווית m הוא היחס בין אורך הניצב מול הזווית לבין אורך היתר במשולש ישר זווית.

$$\sin(m) = \frac{b}{c}$$

m - מבטא את מספר המעלות של הזווית.

תרגיל מתמטי

א. נתון משולש ישר זווית שאורך היתר הוא 100 ס"מ. מהו אורך הניצב מול זווית m אם נתון כי סינוס הזווית שווה ל-0.3.

ב. נתון משולש ישר זווית שבו הניצב מול זווית m הוא 50 ס"מ מהו אורך היתר אם נתון כי: $\sin(m) = 0.70$

(תשובה 3)

מעלות ורדיאנים

ראית כבר כי סינוס של 30° , $\sin(30^\circ)$, שווה ל- .
בוא נפקוד על המחשב להדפיס את ערכו של $\sin(30^\circ)$:
כתוב במחשב: □

PRINT SIN(30)


לאחר שלחצת RETURN - האם המחשב כתב 0.5? ?

לא! הופיע המספר 0.988031623-.

מה קרה פה? האם המחשב "השתגע" ואינו יודע ש- $\sin(30^\circ) = 0.5$?
אל דאגה. המחשב ואתה פשוט לא דיברתם באותה שפה!

המחשב אינו מודד זוויות במעלות, כמוך, אלא ביחידת זווית הנקראת רדיאן.

לפניך 2 זוויות: האחת בת רדיאן אחד והשניה בת מעלה אחת:



1 מעלה

זוויות אפשר למדוד ביחידות שונות בדיוק כפי שניתן למדוד מרחקים וביחידות אורך שונות, כמו: ק"מ, מייל, ס"מ וכדומה. אז איך נאמר למחשב לכתוב $\sin(30^\circ)$? אל דאגה! אנחנו קרבים לפתרון...

כזכור, זווית שטוחה שווה ל- _____ מעלות.
(אינך יודע? - פנה לנספח א בעמוד 45)




כמה רדיאנים יש בזווית שטוחה?

בשרטוט שמשמאל, שרטטנו זווית בנות רדיאן אחד בזו אחר זו. על סמך השרטוט תוכל להעריך

כי בזווית שטוחה (180°) יש:

- בין 2 ל-3 רדיאנים.
 - בין 3 ל-4 רדיאנים.
 - בין 4 ל-5 רדיאנים.
- מה הערכתך?



ובכן, דע לך כי מספר הרדיאנים הנכנסים לתוך זווית שטוחה (180°) שווה ל:

3.14159265

יתכן מאוד כי מספר זה מוכר לך מלימודי ההנדסה בבית הספר. כך למשל, אורך ההיקף של מעגל בעל רדיוס R הוא:
 $2 * \pi * R = 3.14159265 * R * 2$ = אורך ההיקף.
נהוג לציין מספר זה בעזרת האות היוונית π (פאי)

אגב, מאחר שמספר זה כל כך שימושי בהנדסה, עסקו כבר המתמטיקאים בזמן הקדום בחישובו. ארכימדס איש סיראקוז (212-287 לפני הספירה) היה איש המדע הראשון שהראה כיצד מחשבים את הערך של π . ארכימדס מפורסם גם בניסוח חוקי המנוף והציפה של גופים.

זכור!

π (פאי) מציין את המספר _____ שהם מספר הרדיאנים הנכנסים בזווית שטוחה.

עכשיו, כשאתה יודע ש-180 מעלות שוות ל- π רדיאנים, ענה על השאלות הבאות:

זווית של 90° שווה ל- $(\frac{\pi}{3}, \frac{\pi}{4}, \frac{\pi}{2}, \frac{\pi}{6})$ רדיאנים.

(תשובה 4)

זווית של 60 מעלות שווה ל- $(\frac{\pi}{3}, \frac{\pi}{4}, \frac{\pi}{2}, \frac{\pi}{6})$ רדיאנים.

זווית של 45 מעלות שווה ל- _____ רדיאנים.

זווית של 30 מעלות שווה ל- _____ רדיאנים.

(תשובה 5)

סוף סוף הגענו למטרה:

פקוד על המחשב להדפיס את הערך של סינוס 30 מעלות:

PRINT SIN (_____)

אם פעלת בכוון אזי המחשב הדפיס 0.5 ששווה לסינוס של זווית $\frac{\pi}{6}$ בת רדיאנים (המחשב מקצר את הכתיבה של 0.5 ל-0.5).

הערה

אל תשכח את הסוגריים אחרת תקבל SYNTAX ERROR - נסה!

משימה

כתוב תוכנית קצרה אשר בה המחשב שואל אותך (INPUT) את גודלה של זווית במעלות. אחרי שעבית כותב המחשב את הזווית במעלות וגם ברדיאנים.

(קרא למשתנה הזווית במעלות - α ולמשתנה הזווית ברדיאנים - β .)

(תשובה 6)

משימה נוספת

שכלל את התוכנית האחרונה כך, שהתוצאות על המסך יתקבלו כך:

$$m = 60$$

$$z = \pi/3$$

וכן הלאה.

(תשובה 7)

הערה

הרץ את התוכנית ונסה להכניס ל-M את הערך $\pi/3$, למשל. בתגובה

המחשב מודיע: REENTER, כלומר, הוא אינו מוכן לקבל π

לתוך INPUT!

מחק את התוכנית.

SIN(0)


פקוד על המחשב להדפיס את $\text{SIN}(0)$.

מדוע $\text{SIN}(0) = 0$?


עכשיו נסביר זאת.

לפניך משולש ישר-זווית:

$$\text{כזכור, } \text{SIN}(z) = \frac{b}{c}$$



כעת תאר לעצמך שהמשולש הולך ומשתנה, z הולכת וקטנה ומתקרבת ל-0:




אתה רואה כי ככל שהזווית z מתקרבת ל-0, הניצב מול z הולך וקטן ולכן הסינוס של הזווית z הולך ומתקרב ל-0:

$$\text{SIN}(0) = \frac{0}{c} = 0$$

SIN ($\pi/2$)

פקוד על המחשב להדפיס SIN ($\pi/2$). המחשב עדיין זוכר את ערכו של π מהתוכנית הקודמת.


מדוע קיבלת 1? (0.99999999 שווה למעשה ל-1)
תאר לעצמך שהזווית z הולכת וגדלה ומתקרבת ל- $\frac{\pi}{2}$:



ככל שהזווית z הולכת ומתקרבת ל- $\pi/2$ (90°) אורכו של הניצב b מול z הולך ומשתווה לאורכו של היתר c , וכאשר $z = \frac{\pi}{2}$ שווה b ל- c ולכן: $\text{SIN}(\pi/2) = 1$

תרגיל מתמטי


פקוד על המחשב להדפיס את אורך הניצב b :



(תשובה 8)

קוסינוס (COS)

לאחר שהבנת את מושג הסינוס לא יקשה עליך להבין את הקוסינוס:



הקוסינוס של הזווית z הוא היחס

בין הניצב ליד הזווית לבין היתר: $\text{COS}(z) = \frac{a}{c}$

לכל זווית יש ערך קוסינוס ידוע וקבוע!

חזור למשולשים ישרי הזווית בעמוד 7. ☐

מדוד בעזרת סרגל והמחשב את ערך הקוסינוס של זווית בת 30° . ☐

כדאי שתבצע את מדידותיך על המשולש הגדול ביותר כדי לקבל תוצאה מדויקת ככל האפשר.


ועכשיו פקוד על המחשב להדפיס את ערך הקוסינוס של זווית זאת. ☐

תוצאות המדידה הם: $\frac{a3}{c3} = \underline{\hspace{2cm}}$

ועכשיו פקוד על המחשב להדפיס את ערך הקוסינוס של זווית זאת: ☐

PRINT COS ()

האם תוכל "לנחש" מהו הקוסינוס של 60° ? ☐




$$\cos(60^\circ) = \frac{a}{c} = ?$$

פקוד על המחשב להדפיס $\cos(60^\circ)$. ☐

(אל תשכח להפוך לרדיאנים)

קיבלת וזה בדיוק כמו סינוס של 30° . האם זה מפתיע?
תוצאה זאת לא היתה צריכה להפתיע אותך כי הרי הזווית החדה השניה במשולש ישר זווית בו אחת הזוויות היא 30° תהיה 60° :



$$\sin(30^\circ) = \cos(60^\circ) \text{ כי}$$

התוכל לשער מה יהיו ערכי הקוסינוסים במקרים הבאים: ☐

(רמז: חשוב כל הזמן מה קורה $\cos(\emptyset) = \underline{\hspace{2cm}}$

לאורך הניצב ליד הזווית ביחס


ליתר במקרים אלה.) $\cos(\pi/2) = \underline{\hspace{2cm}}$

כדוק עצמך עם המחשב. ☐

תרגילים מתמטיים

1. מה אורך הניצב a אם


$$\cos(z) = 0.37$$



2. פקוד על המחשב להדפיס את

אורך הניצב a ואת אורך

הניצב b :




(תשובה 9)

מתיחת קו על המסך

עליך למתוח על המסך קו באורך 100 מהפיקסל (20, 40) בזווית נטיה של 20 מעלות ביחס לקו האופקי:

זווית זאת (במקרה שלנו 20°)


נקראת בשם "זווית הנטיה של הקו"



(תשובה 10)

חידון זווית

המחשב מגריל זווית בגודל אקראי, בין 0 ל- 90 מעלות ומשרטט אותה על המסך:



(שים לב! הקו האופקי מופיע על המסך באופן קבוע והקו השני מופיע בהתאם לזווית הנטיה שהוגרלה.)

עליך לבחש כמה מעלות יש בזווית המוצגת על המסך.

לאחר שניחשת (INPUT) המחשב מדפיס לך את גודלה האמיתי של הזווית במעלות ולידה את הניחוש שלך.

(תשובה 11)


הרחבת המשחק

שים לב! אתה יכול להרחיב את החידון גם לזוויות גדולות מ- 90° , כל שעליך לעשות הוא לפקוד על המחשב להגדיל זווית בין \emptyset ל- 360° .

יתכן שאתה מופתע מכך שהמחשב מכיר בסינוסים וקוסינוסים של זוויות הגדולות מ- 90° וזאת מהסיבה שמושגים אלה הוגדרו עבור זוויות חדות במשולש ישר זווית. אם אתה מעוניין לדעת מדוע המחשב מכיר בסינוסים וקוסינוסים של זוויות הגדולות מ- 90° אנו ממליצים לך לפנות לנספח 2 בעמוד 48.

תוכנית המניפה

כתוב תוכנית כך שהמחשב יצייר בזה אחר זה 6 קווים באורך 70 מנקודת מוצא קבועה (למשל, $8\emptyset$, $14\emptyset$) כך שזווית הנטיה של הקו הראשון תהיה \emptyset , של השני $2\emptyset$, של השלישי $4\emptyset$ וכך הלאה, עד שישרטט קו שזווית נטיתו $1\emptyset$ מעלות.



(תשובה 12)

ועכשיו שנה את התוכנית כך שהמחשב ישרטט קווים שזווית הנטיה של הראשון שווה ל- \emptyset והאחרון 360° בקפיצות של 3 מעלות.

האם המניפה המתקבלת על המסך קצת פחוסה (אליפטית)?
אם כן, תוכל לקבל מניפה מעגלית מושלמת אם תנקוט באחת מן הפעולות הבאות:

- יתכן שאחד מכפתורי המוביטור (צג) שלך מאפשר תיקון העיוות. בדוק זאת.
- אם לא - עליך להכפיל את הסינוס ב- $6/7$, או במספר קרוב לזה - תלוי במידת העיוות.

שילוב של עיפרון וידאו ומניפה

שנה את התוכנית כך, שתוכל להזיז כרצונך בעזרת "עיפרון הוידאו" את מקומה של נקודת המוצא אשר ממנה נמתחים הקווים ותוך כדי כך המחשב ממשיך ומשרטט כל הזמן קווים בעזרת תוכנית המניפה... (אינך זוכר את תוכנית עיפרון הוידאו? - פנה ליחידה 4 פרק ג.) כדאי גם שאורך הקווים יהיה קצר יותר! (למשל 20 פיקסלים).

(תשובה 13)

ועכשיו כשהתוכנית עובדת נוכל להציע לך גם משחק...



המניפה
מתחילה
להסתובב
משמאל
לארובה.

פינת המשחק: מנקה הארובות

הוסף לתוכנית שלך עוד פקודות כך שהמחשב ישרטט "ארובה" (שני קווים מקבילים), כך שמחצית המרחק ביניהם קטן מאורך הקו שיוצר את המניפה.


(קח למשל מרחק של 30 בין הקווים ו-20 לאורך קו המניפה).

משימתך: להעביר את "מנקה הארובות" (המניפה) דרך הארובה (בעזרת עפרון הוידאו) מבלי שהוא יחרוג "מקירות" (הקווים) הארובה.

הנה לפניך מקרה בו "מנקה הארובות" (המניפה) חרג מ"הארובה":

(יתכן שכדאי לך לעבוד

עם 1 OVER).



כאן ישנה חריגה

מעונין במשחק קשה יותר?

- תן לארובה להיות צרה יותר.
- או תן ל"מנקה הארובות" להסתובב יותר מהר.
- (תן STEP גדול יותר של הזזת)
- צייר ארובה "מפותלת".
- תן הקצבת-זמן.

חזרה לתוכנית המבוא...

כעת אתה בהחלט בשל להבין ולכתוב בעצמך את התוכנית שהובאה בתחילת הפרק שיצרה מעגל:

כתוב תוכנית שתגרום למחשב


לצייר מעגל המורכב מנקודות.

רדיוסו 70 ומרכזו ב-80, 140.

הזווית הנוצרת בין כל שתי

נקודות סמוכות ביחס למרכז

תהיה 4 מעלות.



שאלה

מה עליך לעשות על מנת לקבל מעגל

מושלם כדוגמת זה?

ניתן לחשוב על שתי

דרכים כדי לעשות זאת.

האם אתה יודע מהן?

דרך א:

שנה את הוראת ה-STEP בתוכנית על מנת להשלים את הנקודות למעגל שלם.

(תשובה 14)

דרך ב:

השתמש בהוראת H\$PLOT TO על מנת להשלים את הנקודות למעגל שלם.


(תשובה 15)

איזו מבין שתי הדרכים מהירה יותר?



שעשועי מעגלים

□ כתוב תוכנית שתצייר 5 מעגלים בעלי מרכז משותף, שרדיוסיהם גדלים בקפיצות של 10:



(תשובה 16)

אל תמחק את התוכנית ועבור הלאה:
□ שנה את התוכנית כך, שיתקבלו השרטוטים הבאים:



רמז: תוך כדי שינוי הרדיוס - הזז את מרכזי המעגלים ימינה

(תשובה 17)

במידה הדרושה...

והמשך "לשחק" להנאתך עם התוכנית...

שבלול

כתוב תוכנית שתשרטט

את השבלול הבא:

מתקשה הנה רמז:

חשוב על כך שרדיוס המעגל

הולך וגדל כל הזמן!



(תשובה 18)

אתגר

עכשיו תוכל לפנות ולפתח תוכנית


שתשרטט את השמש הבאה:



(תשובה 19)

משימה 1

ועכשיו פנה לשרטט את הנקודות הבאות:



רמז: הזכר בשילוב של עפרון-וידאו ומניפה.

(תשובה 20)

משימה 2

הפעם לפניך ציור

קשה יותר:



(תשובה 21)

רמז: הפעם המניפה נעה

לאורך מעגל במקום






לאורך קו ישר כמו

במשימה הקודמת.

ולסיום...

נשאיר אותך להתמודד לבדך מול הציורים הבאים:

המשימות אינן פשוטות. אם אתה מתקשה עבור הלאה ואח"כ חזור אליהן.



פרק ב

שעון מחוגים

בניית שעון מחוגים אלקטרוני

זוכר כי ביחידה 2 פיתחת שעון דיגיטלי?

עכשיו, לאחר שאתה שולט בסיבוב ובקוסינוס, תוכל לבנות לעצמך שעון מחוגים.


נבנה אותו בשלבים:

שלב 1

נתחיל דווקא מן המסגרת של השעון.

כתוב תוכנית שתצייר, בגודל

מכסימלי, את מסגרת השעון הבאה:



הערה במעגל החיצוני יש 12 בקודות

עבור 12 שעות, ובמעגל הפנימי - 60 בקודות עבור 60 דקות.

(תשובה 22)

שלב 2

בשלב זה נבנה את מחוג השניות.

הוסף לתוכנית עוד שורות כך, 

שהמחוג יסתובב ללא הפסק בתוך

המעגל הפנימי.

שים לב לבקודות הבאות:



- המחוג צריך להיות ארוך ככל האפשר - אך שלא יפגע בנקודות.

- המחוג צריך להתחיל להסתובב מן השעה 12.

- המחוג צריך להסתובב בכיוון מחוגי השעון.

- המחוג צריך להסתובב ללא הפסק.

- המחוג צריך להסתובב בקפיצות של דקה כלומר - 60 קפיצות

כל סיבוב.

זקוק לעזרה?

יתכן שנתקלת באחת הבעיות הבאות:

- המחוג אינו מתחיל להסתובב מן השעה 12 - תן ל-M ערך התחלתי מתאים.

- המחוג מסתובב הפוך - הקטן את M במקום להגדילו.

(תשובה 23)

האם אתה שולט על קצב ההתקדמות של המחוג? גסה!

שלב 3

בשלב זה נוסיף את מחוג הדקות:

- מחוג הדקות קופץ בדקה בכל פעם שמחוג השניות משלים סיבוב מלא.

- מחוג הדקות קצר יותר ממחוג השניות.

הערה:

יתכן שאת מחוג השניות ציירת בעזרת לולאת GO TO - גם אנחנו, בתשובה, פתרנו זאת בעזרת GO TO.

כעת, לעומת זאת, כאשר רוצים להכביס את מחוג הדקות לתמונה, ורוצים שהוא יתקדם בכל פעם שמחוג השניות משלים סיבוב מלא - יש צורך לבצע את שניהם בעזרת לולאת FOR-NEXT, בדומה לדרך שפתרת זאת בחוברת 2, כאשר בנית שעון דיגיטלי.

הזהר מהבגים האפשריים הבאים:

- מחוג השניות מוחק את מחוג הדקות כאשר הוא עובר מעליו - מחוג הדקות צריך להצטייר כל הזמן מחדש.

- מחוג הדקות מהבהב - אל תעדכן אותו כל הזמן אלא רק כאשר מחוג השניות משלים סיבוב מלא.

- מחוג השניות מצטייר פעמיים בשעה 12 - שנה את אחד המספרים בפקודת ה-FOR.

(תשובה 24)

פרק ג

סינוסואידות

הסינוסואידה

☐ הרץ את התוכנית הבאה במחשב:

```
5  PI = 3.14159265
10 HGR : HCOLOR = 3
20 FOR M=0 TO 360 STEP 3
25 Z=PI/180*M
30 Y=70*SIN (Z)
40 X=M/3
50 HPLOT X, 80-Y
60 NEXT M
```

הצורה המתקבלת על המסך בקראת סינוסואידה. מקור השם של צורה זאת קשור, כמובן, לסינוס המופיע בתוך התוכנית. אם חשקה בפשך לדעת כיצד התקבלה צורה כזאת - פנה לנספח: "הסבר ציורי-סינוסואידות" בעמוד 52.

אבל גם ללא הנספח תוכל להבנות מן הסינוסואידות:

☐ שבה את המספר 70 המופיע בשורה 30 וראה מה מתקבל על המסך.
☐ מה המספר הגדול ביותר שעדיין משאיר את הסינוסואידה בתחומי המסך?

(תשובה 25)

☐ שבה את התוכנית כך, שהמחשב ישאל אותך מה גודל המשתנה (קרא לו - A) שאתה רוצה להציב במקום 70 בשורה 30 - יצייר את הסינוסואידה המתאימה לגודל שהכנסת, ויחזור לשאול אותך לגודלו של A - בלי למחוק את הציור הקודם.

(תשובה 26)

דאג לכך שה-INPUT יופיע בחלון הטקסט בחלק התחתון של המסך.

אמפליטודה (משרעת)


למשתנה A הקובע את גובה הסינוסואידה קוראים בעברית משרעת (או בלועזית אמפליטודה).

☐ נסה להכניס ערכים שליליים ל-A, וראה מה מתקבל.


משימה

☐ שנה את התוכנית כך, שהיא תצייר את הצורות הבאות, באופן

אוטומטי - ולא ב-INPUT:



(תשובה 27)



(תשובה 28)

משימה

איזה שינוי עליך להכניס
לתוכנית כדי שתקבל הצורה
הבאה:



כדי לעשות זאת בצורה אלגנטית ביותר רצוי מאוד שתשתמש בהוראה:
ABS (אבסולוט) – ABSOLUTE, שפירושה: **הערוך המוחלט**
אינך מכיר הוראה זאת? פנה ליחידה 4 עמוד 18.

(תשובה 29)

אורך גל

בנוסף לגובה הסינוסואידה (אמפליטודה) אתה יכול גם לשלוט על
אורך הסינוסואידה (הבקרא גם אורך גל).
שנה את התוכנית כך, שתחזור לתוכנית המקורית בעמוד 24. שנה
את שורה 40:

$$40 \quad X=M/2$$

- והרץ את התוכנית. ☐
- שנה את המספר המופיע בשורה 40 וראה מה מתקבל. ☐
- מהו המספר הגורם לסינוסואידה להשתרע בדיוק לרוחב כל המסך? ☐
- (תשובה 30)
- הכנס משתנה במקום המספר בשורה 40 (קרא למשתנה זה - L). ☐
- שנה את התוכנית כך שתשאל אותך מה גודלו של L - ותצייר את ☐
- הסינוסואידה המתאימה על המסך.

חידה


מה אתה צריך לשנות בשורה 20 ובשורה 40 (בתוכנית המקורית) כדי לקבל סינוסואידה מתמשכת:



(תשובה 31)

משימה


שנה את התוכנית כך שתקבל:



(תשובה 32)

משימה


□ כתוב תוכנית שתצייר את הצורה הבאה:
(עדיף במסך גרפי 8)



(תשובה 33)

משימה

□ נסה להתמודד לבדך עם השרטוטים הבאים:



פרק ד

ציורי ליסאז'

אליפסות

תתפלא להיווכח, כמה ציורים יפים ומיוחדים אפשר לקבל על-ידי שינויים קלים בתוכנית המעגל המוכרת לך:

```
10 HGR : HCOLOR=3
15 PI=3.14159265
20 FOR M=0 TO 360 STEP 3
25 Z=M*PI/180
30 X=140 + 70 * COS (Z)
40 Y=80 - 70 * SIN (Z)
50 HPLOT X, Y
100 NEXT M
```

אינך זקוק, ודאי, להרצת התוכנית כדי להיווכח כי היא מציירת מעגל.

אבל

מה יקרה אם תשנה את שורה 30: ?

```
30 X=140 + 1.5 * 70 * COS (Z)
```

נסה ותיווכח. □

איזה שינוי עליך להכניס בשורה 30 כדי לקבל אליפסה כזו: ?



(תשובה 34)

☐ שנה שוב את שורה 30:

$$30 \text{ X} = 140 + 70 * \cos (Z + \pi/4)$$

☐ מה יתקבל אם תריץ כעת את התוכנית.

☐ הרץ ותיווכח.

☐ איזה שינוי עליך להכניס בשורה 30 כדי לקבל אליפסה כזו:



(תשובה 35)

☐ בצע שינוי קל בתוכנית כך, שהאליפסה תיראה כך:



(תשובה 36)

שינויים מפתיעים

כל השינויים שביצענו עד עכשיו בשורה 30 גרמו למחשב לצייר ציורים הנראים כמו מעגל מעוך - כלומר אליפסה.

אבל

לעולם לא תוכל לבחש מה יתקבל על מסך הטלויזיה שלך לאחר שתכניס את השינוי הבא בשורה 30:

$$30 \text{ X} = 140 + 70 * \cos (3 * Z)$$

☐ הרץ את התוכנית.

צורות אלו, המתקבלות על המסך שלך, נקראות צורות ליסאז'ו, על שמו של המתימטיקאי שגילה ופיתח אותן. כעת הזמן להשתעשע מעט...


שנה את שורות 30 ו-40 במקומות שהראנו לך ותוכל לקבל צורות מפתיעות.

בתשובה 37 תוכל לראות כמה מן השינויים שאנחנו הכנסנו. הרץ אותם וראה מה מתקבל.

הערה:

אם תרצה להבין ביתר פירוט כיצד נוצרים ציורי ליסאז'ו וכך גם להגדיל את השליטה שלך כמה שמתרחש על המסך - פנה ללמוד את נספח ד "תוכנית הדגמה לציורי ליסאז'ו" בעמוד 54.

ולסיום... אולי תוכל לשרטט את הצורות הבאות על המחשב ורבות אחרות.




פרק ה




מצולעים משוכללים

מבוא

האם תופתע לדעת,
שהשרטוט היפה הזה:



התקבל בעזרת אחד המצולעים המשוכללים הבאים:



התוכל לבחש איזה מהמצולעים המשוכללים יצר את הצורה היפה שלמעלה?

מהו מצולע משוכלל?

במצולע משוכלל כל הזוויות שוות ביניהן וכל הצלעות שוות באורכן!

שמות של מצולעים

- מצולע מספר ____ נקרא משולש.
- מצולע מספר ____ נקרא ריבוע.
- מצולע מספר ____ נקרא מחומש.
- מצולע מספר ____ הוא משושה.

בפרק זה תלמד לכתוב תוכנית המשרטטת מצולעים משוכללים כלשהם ובעזרתה תצליח לקבל צורות מגוונות ומענינות כמו זו שבתחילת העמוד.

משימה

עליך לכתוב תוכנית, אשר תגרום למחשב לשרטט קוים ישרים השוים באורכם, אחד בהמשכו של השני, כאשר כל קו נטוי, ביחס לקו האופקי בזווית הגדולה מקודמו ב- m מעלות.

באיור שלפניך אתה רואה

3 קוים שווים באורכם.

- הראשון נטוי בזווית θ ביחס לקו האופקי.
- השני נטוי בזווית m מעלות לקו האופקי.
- השלישי נטוי בזווית $2*m$ מעלות לקו זה וכך הלאה...



שים לב

לפני השרטוט המחשב שואל אותך (INPUT):

- כמה קוים אתה רוצה שיופיעו (קרא למשתנה זה N)?
- מהו האורך של הקווים (קרא למשתנה זה L)?
- מה גודל הזווית M במעלות (בתוכנית הפוך זווית זו לרדיאנים). לאחר שהכנסת את כל הנתונים, משרטט המחשב את הקוים על המסך. בידך כל הכלים לבנות התוכנית - גש לעבודה. אתה מתקשה?

ראה את תוכנית "המניפה" בעמוד 16. ההבדל העיקרי בין התוכניות הוא, שבתוכנית המניפה לאחר שרטוט כל קו, המחשב חזר ל-HPLOT במרכז, ואילו כאן...

(תשובה 38)

חקירת התוכנית

עכשיו, כשהתוכנית נמצאת בזכרונו של המחשב, תוכל לחקור אותה על מנת לקבל מצולעים משוכללים מסוגים שונים:

קבלת ריבוע

□ נסה להזין את המחשב בנתונים מתאימים של N , L ו- m על-מנת לקבל ריבוע על המסך.

אתה יכול, כמובן, "לשחק" עם ערכי m ו- N עד שתקבל במקרה ריבוע - אבל:

אנו ממליצים כי תצייר לך ריבוע על נייר ותחשוב מה צריך להיות ערכם של N ו- m על-מנת לקבל ריבוע על המסך.

(תשובה 39)

קבלת משולש

□ נסה לקבל מן התוכנית הזו משולש משוכלל (משולש שווה צלעות). ושוב - חשוב מה צריך להיות ערכם של N ושל m כדי לקבל משולש שווה צלעות.

(רמז: מתוך לימוד נספח 1 אתה אמור לדעת את התשובה.)

(תשובה 40)

אתגר מחשבתי

❓ האם תוכל, כבר עכשיו, להכניס לתוכנית נתונים כאלה, שהמחשב יצייר על המסך מחומש או משושה?
אם אתה מתקשה - עבור לחידה הבאה:

חידה מתמטית

- כדי לקבל משולש (3 צלעות) הכנסת $N=$ ____, $m=$ ____
- כדי לקבל ריבוע (4 צלעות) הכנסת לתוכנית $N=$ ____, $m=$ ____

דע לך כי יש קשר מתמטי ברור בין מספר הצלעות לבין גודל הזווית m .

האם מתוך המקרים בהם טיפלת (משולש וריבוע) תוכל לנסח את הקשר בין n לבין m ?
לא הצלחת? העזר בשאלה הבאה:

באיזה מהקשרים המתמטיים היית בוחר, כנכון:

א. $m = \frac{360}{n}$ (וברדיאנים: $z = \frac{2\pi}{n}$)

ב. $m = \frac{180}{n}$ (וברדיאנים: $z = \frac{\pi}{n}$)

ג. $m = \frac{n}{360}$ (וברדיאנים: $z = \frac{n}{2\pi}$)

(תשובה 41)

משימה

ועכשיו, לאחר ידיעת הקשר בין n לבין m , הזן את התוכנית שבמחשב בנתונים מתאימים על-מנת שיצייר משושה! (6 צלעות)

(תשובה 42)

משימה

כעת שנה את התוכנית כך:

- המחשב ישאל אותך כמה צלעות (n) יש למצולע אותו אתה רוצה לצייר.

- המחשב נענה מיד לבקשתך ומצייר את המצולע הנדרש!

(תשובה 43)

בעזרת תוכנית זו שרטט מחומש, מצולע בעל 7 צלעות, 8 צלעות וכן הלאה.

אתגר

האם תוכל, בעזרת תוכנית זו, לצייר מעגל?

משימה

□ שינוי קטן בתוכנית יגרום למחשב לצייר "שבלול-משושה":
(שים לב כי L כל הזמן משתנה)



שחק עם התוכנית כך שתקבל
"שבלול-ריבועי", "שבלול-מחומש"
וכן הלאה.


(תשובה 44)

משימה

□ שנה את התוכנית כך,

שתקבל את הצורה הבאה:

? האם לכל המצולעים אותו אורך צלע?



(תשובה 45)

המצולעים המסתובבים

בעזרת תוכנית זו הנמצאת עכשיו בזכרון ניתן להגיע

לצורות מקסימות:

שנה את התוכנית כך:

המחשב צייר על המסך

ריבוע ראשון:

1



המחשב ממשיך ומצייר על המסך

ריבוע שני (מסובב בזווית c

סביב נקודת ההתחלה).

2




המחשב אינו אומר די ומשרטט

ריבוע שלישי, מסובב בזווית c

ביחס לריבוע הקודם:

3



נקודת
סיבוב

וכך הלאה עד שהריבועים משלימים


מעגל שלם (360 מעלות).

(תשובה 46)

שחק עם התוכנית

כעת תוכל לשחק עם התוכנית על מנת לקבל צורות כיך הדמיון הטובה עליך:

- שנה את מספר הצלעות וראה מה קורה על המסך.
 - שנה את זווית הנטייה C.
 - ומה יקרה אם תוך כדי הסיבוב הצלע תלך ותקטן?
- נסה לקבל את הצורות שלמטה (או יפות מהן...):



שכלול נוסף ואתגרי

שנה את התוכנית כך, שנקודת הסיבוב אינה נשארת קבועה, אלא נעה אף היא במעגל ככל שהמצולעים הולכים ומצטטירים. כך תקבל צורות כאלה:



(תשובה 47)


חידת פתיתי השלג

השרטוט הבא התקבל

כתוצאה משינוי קטן

שעשינו בתוכנית.

מתקשה?






הנה רמז: עליך לשנות את התוכנית, כך שיתקבל

"מנקה-ארובות" הנע במעגל.

(תשובה 48)

השלג ממשיך לרדת...


שחק בתוכנית שלך ותוכל לקבל צורות כאלו ורבות אחרות:



פרק ה

צמצמים


לפניך שרטוט מחשב המזכיר צמצם של מצלמה:



עכשיו נבנה שלב אחר שלב תוכנית שתצייר צמצמים.

שלב 1: צורות החזרות על עצמן

האם מתוך התבוננות בשרטוט הצמצם למעלה אתה מזהה צורה מסוימת החוזרת על עצמה פעם אחר פעם? אולי השרטוט הבא יבוא לעזרתך?




אתה רואה כי השרטוט בנוי מריבועים ההולכים וקטנים. פנה לשרטוט המחשב שלמעלה ובסה לזהות את הריבועים.

שלב 2: שרטוט ריבוע

כדי לבנות תוכנית ששרטט ריבועים כאלה, נלך צעד אחר צעד. תחילה נבנה ריבוע, כאשר הקואורדינטות של קודקודיו ידועות.

נטיל עליך את המשימה לשרטט על המסך



את הריבוע הבא:

$$A1=160, B1=150$$

$$A2=230, B2=80$$

$$A3=160, B3=10$$

$$A4=90, B4=80$$

כתוב תוכנית ששרטט את הריבוע, שקודקודיו הם בעלי הקואורדינטות הרשומות למעלה.

דאג לכך שהמחשב יתחיל לשרטט את צלע 1, לאחר מכן את 2 וכך הלאה.

הערה: תן, כתחילת התוכנית, ערכים מחאימים ל-A1, ... A4

ול-B1, ... B4 והשתמש אח"כ רק בשמות המשתנים.

לאחר שהרצת את התוכנית השווה אותה לתשובה 49.

שכלול התוכנית

מאחר שאתה יודע כבר להשתמש במערכים, נטיל עליך את המשימה הבאה:

לפניך תוכנית האמורה לשרטט את הריבוע תוך שימוש במערכים.

לתוכנית יש בג! משימתך היא למצוא את הבג של התוכנית על-מנת

שהיא אכן תבצע את המשימה של שרטוט הריבוע:

```
5 HGR : HCOLOR=3
```

```
10 DIM A(4) : DIM B(4)
```

```
20 FOR I=1 TO 4
```

```
30 READ A(I), B(I)
```

```
40 NEXT I
```

```
50 DATA 160,150,230,80,160,10,90,80
```

```
100 HPLOT A(1), B(1)
```

```
110 FOR I=2 TO 4
```

```
120 HPLOT TO A(I), B(I)
```


```
130 NEXT I
```

האם מצאת את הבג? תקן אותו.

לאחר שתתמודד בסבלנות עם הבעיה פנו לתשובה 50.

שלב 3: מציאת קודקודי הריבוע הפנימי

עכשיו עליך למצוא את הקואורדינטות של קודקודי הריבוע הפנימי:



בשלב זה, נניח שאנחנו רוצים שהקודקוד x_1, y_1 יפול בדיוק באמצע הצלע של הריבוע החיצוני.

במקרה כזה קל מאוד להוכיח (בעזרת סינוסים או בעזרת משולשים דומים - נסה!) כי גם הקטעים $(A(2) - A(1))$ ו- $(B(2) - B(1))$ נחלקים בדיוק באמצע.

$$\begin{aligned} x_1 &= A(1) + ? \\ y_1 &= B(1) + ? \end{aligned} \quad \text{לכן: } \text{?}$$

(תשובה 51)

רוצה להשתכנע שהקטעים נחלקים באמצעיתם? - קח סרגל ומדוד!

פקוד על המחשב לכבות את הפיקסל x_1, y_1 . C

(תשובה 52)

המשך ופקוד על המחשב לכבות את הפיקסלים הנמצאים על אמצע 3 הצלעות הנותרות.

לצורך כך הגדר שני מערכים חדשים:
 $X(?)$, $Y(?)$ מה גודלם?

ותן לאברי המערכים ערכים מתאימים, כמו שחישבת עבור $X1$, $Y1$.
 (תשובה 53)

התבונן היטב במסך - האם כבו כל 4 הפיקסלים הנמצאים במרכזי
 הצלעות? אם כן אפשר להמשיך.

שלב 4: משכפלים את הצורה

כעת עליך למתוח קוים בין הפיקסלים שכיבית כדי ליצור את
 הריבוע הפנימי.

אפשר, כמובן, לכתוב עוד קטע תוכנית שיבצע זאת. אך כבר יש
 בתוכנית קטע שמבצע בדיוק את אותה פעולה - הקטע המצייר את
 הריבוע החיצוני!

כל שעליך לעשות הוא לתת לשני המערכים $A(5)$ ו- $B(5)$ את הערכים
 של המערכים $X(5)$ ו- $Y(5)$.


אל תשכח את הבג שהיה לך עם $A(5)$, $B(5)$ - תקן זאת גם ב-
 $X(5)$, $Y(5)$!
 (תשובה 54)

ובשלב זה אנחנו מקווים שאתה נהנה מהתוצאה המתקבלת על המסך:
 המחשב ממשיך לשרטט ריבועים בזה אחר זה אחד בתוך השני, כך
 שמתקבל צמצם.

☐ שנה את יחס החלוקה מ-0.5 ל-0.2 וראה כיצד זה משפיע על מבנה
 הצמצם. (במקום לחלק ב-2, בשורות 160, 170, יש לחלק ב-5).

☐ הכנס נתונים חדשים כך שהמחשב ישרטט צמצם בצורת מלבן על כל

המסך:




(תשובה 55)

☐ הכנס נקודות כלשהן וראה מה קורה על המסך.

משימה לסיום

קיבלת צמצמים הבנויים ממצולעים משוכללים.
אתה בודאי שם לב כי הצמצמים היפים מתקבלים כאשר ישנן צורות
סימטריות כמו ריבוע ומלבן.
בפרק ה, פיתחת תוכנית לקבלת מצולע משוכלל בעל N צלעות.

המשימה האחרונה ליחידת לימוד זו היא לקבל צמצם הבנוי
ממצולעים משוכללים.
עליך לשנות את התוכנית כך, שהמחשב ישאל אותך מכמה צלעות, N ,
אתה מעוניין שיהיה בנוי המצולע, וכן את אורכה של צלע המצולע -
והמחשב ישרטט באופן אוטומטי את הצמצם:



הפעם לא ניתן לך רמזים.
התמודד בסבלנות עם הבגים ובנה את התוכנית בשלמותה!
ואולי תוכל לשרטט גם את רוח הטורנדו הבאה?



נספח א

זוויות ומשולשים

בנספח זה נלמד אותך מספר מושגי יסוד הקשורים בזוויות ובמשולשים. ניגש ישר לעבודה...

זוויות

איזו מבין הזוויות הבאות היא זווית חדה, ישרה, קהה ושטוחה?




- לכמה מעלות שווה זווית ישרה? (180, 90, 45)
- לכמה מעלות שווה זווית שטוחה? (180, 90, 45)

אם אתה מתקשה לענות סימן כי אינך יודע ששני ישרים מאונכים

החותכים זה את זה יוצרים 4

זוויות שוות שכל אחת מהן שווה

ל- 90° :



שים לב כי זווית שטוחה מורכבת

למעשה משתי זוויות ישרות:



"שחק" את תפקיד המחשב. איזה ערך (1 או 0), היית מדפיס לכל

אחד מהביטויים הלוגיים הבאים: (אינך זוכר מהו ביטוי לוגי?


(פנה ליחידה 4 - עמוד 24)

1. PRINT (זווית ישרה > זווית קהה) ?
2. PRINT (זווית חדה < זווית ישרה) ?
3. PRINT (זווית ישרה > זווית שטוחה) ?
4. PRINT (זווית קהה > זווית שטוחה) ?

(תשובה 56)

זוויות קודקודיות

לפניך שני ישרים החותכים זה את זה. זווית אחת שווה ל- 25° .



למה שווה לדעתך הזווית השנייה m ?

שתי זוויות כאלה נקראות בשם

זוויות קודקודיות והן שוות

ולכן זווית m שווה אף היא ל- 25° .

האם אתה מזהה עוד זוויות קודקודיות באיור זה?

זוויות מתאימות

לפניך שני ישרים

מקבילים k ו- l .


(קווים מקבילים הם קווים

שלעולם לא נפגשים כדוגמת

פסי מסילת הברזל)

קו שלישי, i , חותך את שני

הקווים המקבילים.



באיור מצוינות שתי זוויות. הזווית העליונה שווה ל- 50° . למה


שווה לדעתך הזווית השנייה, m ?

שתי זוויות כאלה נקראות בשם: זוויות מתאימות והן אכן שוות


זו לזו ולכן גם זווית m שווה ל- 50 מעלות.

התבונן באיורים הבאים וקבע על סמך מה שלמדת זה עתה על זוויות


קודקודיות ומתאימות את ערך הזווית m בכל מקרה ומקרה:




$m =$ _____



$m =$ _____



$m =$ _____



זוויות מתחלפות


בתרגיל הקודם ראית כי שתי
זוויות כדוגמת אלה המופיעות
באיור הבא, הן m -ו- n
(שוות, שונות)

זוויות אלה נקראות בשם זוויות מתחלפות בין ישרים מקבילים.
זהה עוד צמד של זוויות מתחלפות באיור.


סכום זוויות במשולש

קרוב לוודאי שאתה יודע כי סכום הזוויות במשולש שווה
ל- מעלות.

על סמך הידע שצברת בנספח תוכל להבין מדוע סכום הזוויות
במשולש שווה ל-180 מעלות.



הנה לפניך משולש כלשהו:
שלוש הזוויות של המשולש נקראות
 m ו- n ו- p :



ועכשיו נעביר קו מקביל (b)
לצלע a, העובר דרך קודקוד
זווית p:

סכום שלוש הזוויות העליונות

$$s + p + t = \underline{\hspace{2cm}}$$

(זכור כי הן מהוות זווית שטוחה.)

אבל זווית s שווה לזווית n (במשולש). (b מקביל ל-a).
וזווית t שווה לזווית m (במשולש).

ולכן גם סכום הזוויות במשולש שווה ל-180 מעלות.

משולש שווה צלעות


לפניך משולש שצלעותיו שוות.

ערכה של כל זווית במשולש


הוא: $(60, 120)$.

זווית m שווה ל- $(120, 60)$.

(לזווית m קוראים בשם זווית חיצונית למשולש.)




משולש שווה שוקיים




לפניך משולש ששתיים מצלעותיו שוות. (b ו-c). ערך הזווית בין שתי צלעות אלה הוא 50 מעלות. מה ערכן של שתי הזוויות הנותרות במשולש: ____ (70, 65, 130) ?

משולש ישר זווית



משולש ישר זווית הוא משולש שאחת מזוויותיו שווה ל-90 מעלות. לכמה שווה זווית m במשולש ישר הזווית המופיע כאן: שים לב!


במשולש ישר זווית שתיים מן הצלעות ניצבות אחת לשניה. במשולש המשורטט, ו- ____ (a, b, c) הם שני הניצבים במשולש. הצלע השלישית, הנמצאת מול הזווית הישרה, נקראת יתר. צלע ____ (a, b, c) היא היתר במשולש ישר הזווית. (תשובה 57)



כמה משולשים ישרי זווית אתה מזהה באיור הבא? ____ האם לכל המשולשים שאתה מזהה יש אותן הזוויות? ____ (כן, לא) (למשל: האם זווית z שווה לזווית m?)

סיכום: בנספח זה למדת להכיר כמה מושגים בסיסיים הקשורים בזוויות כגון: שמות של זוויות, זוויות קודקודיות, זוויות מתאימות, זוויות מתחלפות וראית כי בעזרת מושגים אלה ניתן להבין מדוע סכום הזוויות במשולש שווה ל-180 מעלות. כמו כן הכרת את המושגים הקשורים במשולש ישר זווית.

סינוסים וקוסינוסים של זוויות הגדולות מ-90 מעלות


$$\sin(z) = \frac{?}{?} (b, r)$$

אתה רואה, כי סינוס הזווית הוא:

A circle with a horizontal diameter. A radius is drawn from the center to the upper-left quadrant, labeled r . The angle between the radius and the horizontal diameter is labeled z .


הזווית נמדדת תמיד בין
הקוטר האופקי לבין הרדיוס -
בכיוון המנוגד לכיוון מחוגי-
השעון.

$$\sin(z) =$$

48

ועכשיו נגלה לך כי הזווית z (בעמוד הקודם) שווה ל- 135° .
 פקוד על המחשב להדפיס את ערך הסינוס של 135° .
 (זכור להפוך את המעלות לרדיאנים)

ערך הסינוס מהמחשב הוא: _____
 האם התוצאות דומות? (אם ישנן סטיות קלות בין התוצאות
 כנראה שלא דייקת מספיק במדידות עם הסרגל.)



ומה יהיה ערך הסינוס של זווית
 הגדולה אפילו מ- 180 מעלות,
 כמו במקרה הבא?
 שוב, כמו בכל המקרים הקודמים
 הסינוס יהיה שווה ליחס בין הניצב
 b , הנמתח אל הקוטר האופקי לבין
 הרדיוס.

האם אתה יכול "לנחש" מה יהיה
 ערך הסינוס של זווית בת 225 מעלות? (רמז: יש לזה קשר
 לתוצאות הקודמות.)


ועכשיו פקוד על המחשב להדפיס את ערך הסינוס של זווית בת
 225 מעלות: $\text{SIN}(\text{PI}/4*5) =$ _____
 $\text{PI}=3.14159265$
 אם פעלת נכון קיבלת ערך שלילי!

הסיבה לכך היא זאת: כאשר הניצב b נמצא מתחת לקוטר האופקי
 הוא נחשב לשלילי ולכן גם ערך הסינוס הוא שלילי (הרדיוס
 אינו משנה את סימנו).

איזה מבין הסינוסים של הזוויות הבאות יהיה לדעתך בעל ערך
 שלילי? 120 , 260 , 300 , 150 מעלות.

בדוק עצמך בעזרת התוכנית שמדפיסה את ערך הסינוס של
 זווית.


ומה בקשר לזוויות הגדולות אף מ-360 מעלות?
גם לגביהן חל אותו כלל - כפי שמודגש במעגל הבא:



? סינוס של איזו זווית חדה יהיה שווה לסינוס של זווית בת 405° ?
חשוב לפני שתבדוק עצמך בעזרת המחשב!

? איזה מבין הסינוסים של הזוויות הבאות יהיה חיובי ואיזה שלילי 460, 570, 660, 750.
לאחר שחשבת - בדוק עצמך בעזרת המחשב.


ומה לגבי קוסינוסים של זוויות הגדולות מ- 90° ?
נתבונן תחילה במקרה הידוע של זווית חדה:



אתה רואה כי קוסינוס הזווית z הוא היחס שבין אורך הניצב a לבין היתר r.

$$\cos(z) = \frac{a}{r}$$

? וכיצד תוכל למדוד עם סרגל ומחשב את הקוסינוס של זווית כאשר היא גדולה מ- 90° - כמו במקרה הבא:



עליך למתוח תחילה את הניצב b אל הקוטר האופקי ואז למדוד את אורך הניצב a ולחשב את היחס $\frac{a}{r}$

דע לך, כי גם הקוסינוס עשוי לקבל ערך שלילי.

רוצה לדעת עבור אילו זוויות?

כתוב תוכנית שתשאל אותך ב-INPUT באיזה זווית אתה מעוניין לבדוק את הקוסינוס ואז היא תדפיס על המסך את הקוסינוס והזווית.


אם חקרת מספר רב של זוויות בודאי הגעת למסקנה, כי כאשר הניצב a נמצא משמאל לקוטר האנכי, ערך הקוסינוס של הזווית הוא שלילי:



? כאשר המחשב יבצע את הפקודה הבאה:

```
PI=3.1415:HGR:HPLLOT 140,80 TO  
140+70*COS (PI/180*250), 80-70*SIN  
(PI/180*250)
```

איזה מבין הקווים הבאים יופיע על המסך?



בדוק עצמך במחשב.


ועכשיו אתה אמור להבין במדויק כיצד התקבלה תמונת המניפה בעמוד 16 גם עבור זוויות הגדולות מ-90 מעלות.

נספח ג


הסבר ציורי סינוסואידות

כיצד מתקבלת סינוסואידה על המסך?

כדי להבין זאת נעזר במעגל הבא:



על הציור הבא העלינו שלושה אנכים השוים באורכם לשלושת האנכים המורדים אל הקוטר האופקי שבמעגל:



- הורד עוד אנכים מנקודות שונות על המעגל.
- מדוד את אורכם בעזרת הסרגל ורשום אותם על הציור למעלה.
(רצוי שתמדוד את אורך האנכים במילימטרים)

האם קצותיהם של האנכים מתחילים ליצור סינוסואידה?
כיצד תחשב את אורך האנכים, במחשב, בעזרת הרדיוס והזווית?

?

(תשובה 60)

משימה

כתוב תוכנית, אשר בעזרתה תוכל לקבל את שאר האורכים מבלי למדוד עם סרגל:


- אתה מכניס ב-INPUT את הזווית - והמחשב נותן לך את אורך האנך.

קנה-מידה

בודאי שמת לב לעובדה ש"פרסנו" את הזוויות לאורך ציר ישר. במקרה שלנו כל 5 מ"מ לאורך הציר מייצגים זווית של 15° . 360° (כלומר 2π רדיאנים) נפרסו לאורך 120 מילימטר. לפניך כמה הגדרות:

אורך הסינוסואידה

זהו האורך מתחילת הסינוסואידה ועד סופה.



האורך המכסימלי של הסינוסואידה הוא _____ ($120, 280$).
משבצות. (המקצוענים קוראים לאורך הסינוסואידה בשם: אורך גל)

גובה הסינוסואידה

המרחק האנכי המכסימלי מהציר נקרא בשם: גובה הסינוסואידה.

- כאשר הזווית שווה ל: _____ אז הגובה הוא מכסימלי.
(המקצוענים קוראים לגובה הסינוסואידה בשם: אמפליטודה)

משימה

שנה את התוכנית כך, שתכניס ב-INPUT את אורך הסינוסואידה (L) ואת גובה הסינוסואידה (A) - והמחשב ישרטט באופן אוטומטי את הסינוסואידה.

נספח ד

תוכנית הדגמה לציור ליסאז'ו

בפרק "ציורי ליסאז'ו" הצלחת לקבל צורות מרהיבות על המסך על ידי הכנסת שינויים קטנים בתוכנית המציירת מעגל בעזרת סינוס וקוסינוס.


בנספח זה נפתח תוכנית הדגמה, אשר בעזרתה תוכל גם להבין מדוע התקבלו ציורי ליסאז'ו על המסך.

תוכנית ההדגמה תתבסס על סיבוב מחוגים, כמו אלו שבשעון המחוגים מפרק ב'. נפתח אותה שלב אחר שלב:

שלב 1

כתוב תוכנית שתצייר שני מחוגים באופן הבא:

- המחשב שואל אותך, ב-INPUT, באיזו זווית (m) ביחס לקו האופקי לצייר את המחוג הקטן.



- לאחר מכן המחשב משרטט באופן אוטומטי את המחוג הקצר (באורך 40), ומיד לאחר מכן את המחוג

הארוך פי שניים מהקצר, כאשר הזווית ביניהם קבועה ושווה ל- $\frac{\pi}{4}$ רדיאנים (____ מעלות). לזווית שבין שני המחוגים קוראים "הפרש מופע".

הערה: לפני שאתה פונה לתשובה תוכל לבדוק אם התוכנית עובדת - הכנס m כאלו, שתדע מראש מה צריכה להיות התוצאה - עוד לפני שהתוכנית רצה.

לדוגמא: שאל עצמך כיצד יראו המחוגים על המסך אם תתן ל- m להיות שווה ל- 0° או 45° .

חזור והרץ את התוכנית וראה כיצד המחוגים מסתדרים בזוויות כלשהן.

(תשובה 61)

שלב 2

הוסף שורה לתוכנית כך,
שלאחר שרטוט המחוגים
המחשב ידליק פיקסל
במקום הבא:



הרץ את התוכנית על זוויות m שונות.

(תשובה 62)

שלב 3

לאחר סיום הופעת הפיקסל המחשב ממתיך - לאחר שתלחץ על מקש
כלשהו - המחוגים ימחקו (אך לא הנקודה) והמחשב ישאל אותך
לזווית m נוספת. לאחר שתזין אותו בזווית הנוספת המחשב ישרטט
את המחוגים ופיקסל נוסף יופיע.

הכנס ב-INPUT נתוני- m נוספים.

(הכנס m מ- 0 ועד 360 מעלות)

איזו צורה מתקבלת על המסך?

אינך בטוח? - המשך להכניס עוד זוויות.

אנחנו מתארים לעצמנו כי אתה מתייגע להכניס בכל פעם m ב-INPUT.
הכנס לולאת FOR-NEXT כזאת, שתסרוק את כל זוויות m מ- 0 עד 360 ,
בקפיצות של 10 מעלות.

כל זמן שתלחץ על מקש המחשב ישרטט פיקסלים. אם תשחרר את המקש -
המחשב יעצר.

(כדאי לך ללחוץ על מקש הרווח, דווקא.)

? מה יקרה לאליפסה אם תשנה את הזווית הקבועה שבין שני המחוגים. במקום $\frac{\pi}{4}$ (45°) היא תהיה $\frac{\pi}{6}$ (____ מעלות).

□ שנה את התוכנית כך, שהמחשב ישאל אותך ב-INPUT לזווית שבין המחוגים ואז הוא יתחיל לשרטט באופן אוטומטי.

קרא למשתנה זה P.

? איזו אליפסה מתקבלת, כאשר הזווית בין המחוגים שווה ל- 0° ?

? מה קורה כאשר הזווית בין המחוגים שווה ל- 90° ?

כאשר הפרש המופע היה $P = \pi/4$ קיבלנו אליפסה מוטית, וכאשר הפרש

המופע היה $P = ______$ קיבלנו אליפסה ישרה כמו זו שבעמוד 29.

? מה יקרה לאליפסה, אם תשנה את תפקידי המחוגים. כלומר, הארוך

יקבע את הקואורדינטה האופקית והקצר את הקואורדינטה האנכית? -

נסה!

מהירות הסיבוב של המחוגים

עד כה דאגנו לכך ששני המחוגים יסתובבו באותה מהירות.

מה יקרה אם, לדוגמא, ניתן למחוג הארוך לנוע פי שניים יותר

מהר מהקצר? (כלומר, שבזמן שזווית הנטיה של המחוג הקצר גדלה

ב- m מעלות, תגדל זווית הנטיה של המחוג הארוך ב- $2*m$ מעלות.)

□ הכנס שינוי כזה, שהמחשב ישאל אותך פי כמה יותר מהר אתה רוצה

שהמחוג הארוך ינוע ביחס לקצר - ובהתאם לכך הוא ישרטט את

הפיקסלים.

בדוק מה מתקבל עבור מהירויות שונות.

נסכם: ניתן להבין את היווצרות האליפסה בעזרת שני המחוגים:
מחוג אחד קובע את הקואורדינטה האופקית (X) של הפיקסל.
והמחוג השני קובע את הקואורדינטה האנכית (Y) של הפיקסל.
במקרה שלנו המחוג הקצר קבע את הקואורדינטה _____ (האופקית,
האנכית) והמחוג הארוך את הקואורדינטה ה _____.

ראינו כי גודל הזווית שבין המחוגים קובע את צורת האליפסה.
נהוג לקרוא לזווית שבין שני המחוגים בשם הפרש המופע שבין
המחוגים.

לסיום

על מנת לקבל את ציורי ליסאז'ו אין צורך, כמובן, לשרטט את
המחוגים עצמם - אלא רק את הפיקסלים. אם תמחק מהתוכנית את
השורות המשרטטות ומוחקות את המחוגים עצמם - תחזור לתוכנית
המקורית, שציירה את צורות ליסאז'ו בפרק ד'.

עכשיו אתה אמור לא רק לצייר, אלא גם להבין את אופן קבלת
ציורי ליסאז'ו!

תשובות

תשובה 1

כל המספרים צריכים להיות, אם מדדת היטב, קרובים או זהים ל-0.5.

תשובה 2

אם, במשולש ישר זווית, אחת הזוויות היא בת 30° - הרי הזווית הנותרת היא בת 60° : $30^\circ + 60^\circ + 90^\circ = 180^\circ$. לכן $\sin(60^\circ) = \frac{a}{c} = 0.87$

תשובה 3

א. אם אורך הניצב הוא X , אז: $\frac{X}{100} = 0.3$

ולכן $X = 30$

ב. הפעם אורך היתר הוא X ולכן $\frac{50}{X} = 0.7$

כלומר: $X = \frac{50}{0.7} = 71.4$

תשובה 4

זווית של 90° שווה ל- $\pi/2$ רדיאנים.

תשובה 5

זווית של 60° שווה ל- $\pi/3$ רדיאנים.

זווית של 45° שווה ל- $\pi/4$ רדיאנים.

זווית של 30° שווה ל- $\pi/6$ רדיאנים.

תשובה 6

5 $\pi = 3.14159265$

10 INPUT "M=";M

20 $Z = M * \pi / 180$

30 PRINT "Z=";Z

40 GO TO 10

תשובה 7

יש לשנות:

20 $Z = 180 / M$

30 PRINT "Z= π ";Z

תשובה 8

PRINT $50 * \sin(35 * \pi / 180) = 28.6788218$

תשובה 9

$\frac{a}{100} = \cos(z) = 0.37$.1

$a = 37$

ולכן:

$$\frac{b}{30} = \sin(40^\circ)$$

2. כדי למצוא את b:

$$b = 30 * \sin(40^\circ) = 30 * \sin(40 * \pi / 180) = 19.28$$

$$\frac{a}{30} = \cos(40^\circ)$$

וכדי למצוא את a:

$$a = 30 * \cos(40^\circ) = 22.98$$

תשובה 10

```
5  PI = 3.14159265
10 HGR : HCOLOR = 3
20 HPLOT 140, 80
30 Z=20 * PI/180
40 A=100 * COS (Z)
50 B=100 * SIN (Z)
60 HPLOT TO 140+A, 80-B
```

תשובה 11

```
5  PI = 3.14159265
10 HGR : HCOLOR = 3
20 HPLOT 100, 100 TO 220, 10
30 M = INT(RND(1)*90)
40 Z = M*PI/180
50 A=100*COS (Z)
60 B=100*SIN (Z)
70 HPLOT 100, 100 TO 100+A, 100-B
80 INPUT N
90 PRINT "N="; N
100 PRINT "M="; M
110 GET A$: GO TO 10
```

תשובה 12

```
5  PI = 3.14159265
10 HGR : HCOLOR = 3
20 FOR M=0 TO 100 STEP 20
30 HPLOT 140, 80
40 Z=M*PI/180
50 HPLOT TO 140+70*COS (Z), 80-70*SIN (Z)
```

תשובה 13

```
15 X=140 : Y=80
30 HPLOT X, Y
50 HPLOT TO X+70*COS (Z), Y-70*SIN (Z)
55 GET A$
56 X=X+3*(A$="K") - 3*(A$="J")
58 Y=Y+3*(A$="M") - 3*(A$="I")
100 GO TO 20
```

תשובה 14

```
5 PI = 3.14159265
10 HGR : HCOLOR = 3
20 FOR M=0 TO 360 STEP 1
30 Z=M*PI/180
40 HPLOT 140+70*COS(Z), 80-70*SIN(Z)
50 NEXT M
```

תשובה 15

```
15 HPLOT 210, 80
20 FOR M=0 TO 360 STEP 5
40 HPLOT TO 140+70*COS(Z), 80-70*SIN(Z)
```

יש להוסיף או לשנות:

תשובה 16

```
12 FOR R=10 TO 50 STEP 10
15 HPLOT 140+R, 80
40 HPLOT TO 140+R*COS(Z), 80+R*SIN(Z)
70 NEXT R
```

יש להוסיף או לשנות:

תשובה 17

```
12 FOR R=10 TO 70 STEP 10
15 HPLOT 140+A+R, 80
40 HPLOT TO 140+A+R*COS(Z), 80+R*SIN(Z)
60 A=A+10
```

יש להוסיף או לשנות:

תשובה 18

```
5  PI = 3.14159265
10 HGR : HCOLOR = 3
15 HPLOT 140, 80
20 FOR M=0 TO 360 STEP 5
40 Z=M*PI/180
50 HPLOT TO 140+R*COS(Z), 80-R*SIN(Z)
55 R=R+0.1
60 NEXT M
```

תשובה 19

```
5  PI = 3.14159265
10 HGR : HCOLOR = 3
20 FOR M=0 TO 360 STEP 5
30 Z=M*PI/180
40 HPLOT 140+30*COS(Z), 80-30*SIN(Z)
   TO 140+70*COS(Z), 80-70*SIN(Z)
50 NEXT M
```

תשובה 20

```
5  PI = 3.14159265
10 HGR : HCOLOR = 3
15 X = 40
20 FOR M=0 TO 360*8 STEP 5
30 Z=M*PI/180
40 HPLOT X+30*COS(Z), 80-30*SIN(Z)
45 X=X+0.3
50 NEXT M
```

תשובה 21

```
5  PI = 3.14159265
10 HGR : HCOLOR = 3
20 FOR M=0 TO 360*6 STEP 5
```

המשך בעמוד הבא...

המשך תשובה 21

```
30 Z=M*PI/180
35 X=140+50*COS (Z/6)
38 Y=80+50*SIN (Z/6)
40 HPLOT X+30*COS(Z), Y-30*SIN(Z)
50 NEXT M
```

תשובה 22

```
5  PI = 3.14159265
10 HGR : HCOLOR = 3
20 FOR M=0 TO 360 STEP 6
30 Z=M*PI/180
40 HPLOT 140+70*COS(Z), 80-70*SIN(Z)
50 NEXT M
60 FOR M=0 TO 360 STEP 30
70 Z=M*PI/180
80 HPLOT 140+75*COS(Z), 80-75*SIN(Z)
90 NEXT M
```

תשובה 23

יש להוסיף:

```
100 M=90
110 Z=M*PI/180
120 HPLOT 140, 80 TO 140+68*COS(Z), 80-68*SIN(Z)
130 FOR T=1 TO 100 : NEXT T
140 HCOLOR = 0
150 HPLOT 140, 80 TO 140+68*COS(Z), 80-68*SIN(Z)
160 M=M-6
170 HCOLOR=3
180 GO TO 110
```


תשובה 24

95 I=90
97 D=I*PI/180
100 FOR M=90 TO -264 STEP -6
125 HPLOT 140, 80 TO 140+50*COS(D), 80-50*SIN(D)
160 HCOLOR = 3
170 NEXT M
175 HCOLOR=0:HPLOT 140, 80 TO 140+50*COS(D),
80-50*SIN(D)
177 I=I-6 : HCOLOR=3
180 GO TO 97

תשובה 25

המספר הגדול ביותר שהמחשב "מרשה" להציב בשורה 30 הוא - 80.

תשובה 26

15 VTAB 24:INPUT "A="; A
30 Y=A*SIN(Z)
70 GO TO 15

תשובה 27

15 FOR A=0 TO 80 STEP 10
70 NEXT A

תשובה 28

15 FOR A=-80 TO 80 STEP 10
יש לשנות:

תשובה 29

15 FOR A=0 TO 80 STEP 10
30 Y=ABS (A*SIN(Z))
יש להוסיף או לשנות:

תשובה 30

40 X=M/(360/279)

20 FOR M=0 TO 720 STEP 3

15 FOR L=5 TO 2 STEP -0.5

70 NEXT L

5 PI = 3.14159265

10 HGR : HCOLOR = 3

20 FOR M=0 TO 720 STEP 6

25 Z=M*PI/180

30 Y=70*SIN(Z)

40 X=M/3

50 HPLOT 0, 80 TO X, 80-Y

60 NEXT M

30 X=140+0.5*70*COS(Z)

30 X=140+70*COS (Z-PI/4)

50 HPLOT X, Y TO X+10, Y

30 X=140+70*COS (5*Z)

30 X=140+70*COS (3*Z+PI/2)

30 X=140+70*COS (3*Z+PI/4)

30 X=140+70*COS (Z/2)

10 X1=100 : Y1=140

15 PI = 3.14159265

20 HOME : VTAB 21

המשך בעמוד הבא...

תשובה 31

יש לשנות:

תשובה 32

יש לשנות:

תשובה 33

תשובה 34

תשובה 35

תשובה 36

תשובה 37

תשובה 38


```

30 INPUT "N="; N
40 INPUT "L="; L
50 INPUT "M="; M
60 Z=M*PI/180
70 HGR : HCOLOR = 3
80 HPLOT X1, Y1
90 FOR I=1 TO N
100 X=X1+L*COS (Z1)
110 Y=Y1-L*SIN (Z1)
120 HPLOT TO X, Y
125 Z1=Z1+Z : X1=X : Y1=Y
130 NEXT I
140 GET A$
150 GO TO 20

```

תשובה 39

כדי לקבל ריבוע:

M=90 N=4

תשובה 40

כדי לקבל משולש:

M=120 N=3

תשובה 41

תשובה א היא הנכונה

תשובה 42

יש להכניס ב-INPUT:

M=60 N=6

תשובה 43

יש לשנות:

למחוק 50

60 Z= $\underbrace{(360/N)}_M$ * PI/180

תשובה 44

יש להוסיף או לשנות:

90 FOR I=1 TO 5*N

127 L=L-1

תשובה 45

יש להוסיף או לשנות, לעומת "השכלול"

```
20 HGR : HCOLOR = 3
30 L=35
40 FOR N=3 TO 12
70   למחוק
90   FOR I=1 TO N
127   למחוק
140 NEXT N
150   למחוק
```

תשובה 46

```
10 X=140 : Y=80
20 HGR : HCOLOR=3
30 PI = 3.14159265
40 HPLOT X, Y
50 N=4
60 L=40
70 M=360/N
80 FOR C=0 TO 360 STEP 30
90   FOR I=1 TO N
100   Z=PI/180*(M*(I-1)+C)
110   X=X+L*COS(Z)
120   Y=Y-L*SIN(Z)
130   HPLOT TO X, Y
140   NEXT
150 NEXT
```


תשובה 47

יש להוסיף או לשנות:

```
10 L1=25
40 למחוק
50 N=6
60 L=30
80 FOR C=0 TO 360 STEP 15
83 X=140 + L1*COS (C*PI/180)
85 Y=80 - L1*SIN (C*PI/180)
87 HPLOT X, Y
```

תשובה 48

יש להוסיף או לשנות:

```
90 למחוק
82 FOR I=1 TO N
```

כלומר העברנו את שורה 90 מיד אחרי 80.

תשובה 49

```
10 HGR : HCOLOR = 3
20 A1=160 : A2=230 : A3=160 : A4=90
30 B1=150 : B2=80 : B3=10 : B4=80
```

תשובה 50

יש להוסיף או לשנות:

```
10 DIM A(5), B(5)
60 A(5)=A(1) : B(5)=B(1)
110 FOR I=2 TO 5
```

תשובה 51

```
X1 = A(1) + (A(2)-A(1))/2
Y1 = B(1) + (B(2)-B(1))/2
```

תשובה 52

```
140 FOR T=1 TO 300 : NEXT T
150 X1 = A(1) + (A(2)-A(1))/2
160 Y1 = B(1) + (B(2)-B(1))/2
170 HCOLOR=0 : HPLOT X1, Y1
```

תשובה 53

```

10 DIM A(5), B(5), X(5), Y(5)
15 FOR I=1 TO 4
16 X(I)=A(I)+(A(I+1)-A(I))/2
17 Y(I)=B(I)+(B(I+1)-B(I))/2
18 HCOLOR=0 : HPLOT X(I), Y(I)
19 NEXT I

```

יש להוסיף או לשנות:

תשובה 54

```

18 A(I)=X(I) : B(I)=Y(I)
20 GO TO 60

```

יש להוסיף או לשנות:

תשובה 55

```

50 DATA 0, 159, 279, 159, 279, 0, 0, 0

```

תשובה 56

- | | |
|----|---|
| 1. | 1 |
| 2. | 0 |
| 3. | 1 |
| 4. | 1 |

תשובה 57

סכום הזוויות במשולש הוא 180° . זווית אחת היא בת 90° לכן סכום שתי הבותרות הוא 90° . ולכן:


$$m = 90^\circ - 37^\circ = 53^\circ$$

תשובה 58

$$\sin(z) = \frac{b}{r}$$

תשובה 59

$$\sin(z) = \frac{b}{r}$$



תשובה 60

עבור זווית 75 מעלות, למשל, יש לכתוב

```
PI=3.1415 : PRINT 25*SIN (75*PI/180)
```

התשובה מתקבלת במילימטרים

תשובה 61

```
10 PI = 3.14159265
20 HGR : HCOLOR=3 : HOME
30 VTAB 22 : INPUT "M="; M
40 Z = M*PI/180
50 HPLOT 140, 80, TO 140+40*COS(Z),
   80-40*SIN(Z)
60 HPLOT 140, 80 TO 140+75*COS(Z+PI/4),
   80-75*SIN(Z+PI/4)
70 GO TO 30
```

תשובה 62

יש להוסיף:

```
65 HPLOT 140+40*COS(Z)
   80-75*SIN(Z+PI/4)
```


שאלת חזרה לסכום

לפניך רשימת מושגים שלמדת ביחידה זו.

עבור על כל אחד מהם ובדוק אם אתה זוכר אותם:

(תוכל לרענן את זכרוןך בעזרת מספרי העמודים המופיעים בסוגריים ליד כל מושג.)

סינוס (7)	סינוסורידה (24)
רדיאן (9)	אמפליטודה (25)
π (10)	אורך גל (26)
$\sin(\theta)$ (12)	ציורי ליסאז'יו (30)
$\sin(\pi/2)$ (13)	הפרש מופע (54)
$\sin(Z), Z > \pi/2$ (48)	מהירות סיבוב (56)
קוסינוס (13)	מצולע משוכלל (32)

ולסיום...

הגעת לסופה של יחידת הלימוד החמישית של מחשב.

בחוברת זאת שמנו לעצמנו כמטרה ללמד אותך את יסודות

הטריגונומטריה על-מנת לקבל שרטוטי מחשב מרהיבים על המסך.

דע לך כי אין זאת החוברת האחרונה של סדרת מחשבת עבור

ה-APPLE.

בחוברת 6, הנקראת "מסע אל זכרוננו של המחשב" נחדור אל זכרוננו

של המחשב בעזרת פקודות ה-POKE וה-PEEK.

חוברת זאת מהווה בסיס ללימוד שפת המכונה של מחשב ה-APPLE.

עד כה יצאו סדרות "מחשבת" עבור:

● מחשב 64 COMMODORE

(3 החוברות הראשונות מתאימות גם ל-VIC-20)

● מחשבי SINCLAIR (SPECTRUM, ZX-81)

● מחשבי APPLE (II, II+, IIe, IIC)

● מחשבי ATARI (800XL, 600XL, 130XE)

● מחשב SPECTRAVIDEO

● מחשב AMSTRAD

● מחשב IBM-PC

● מחשב 128 COMMODORE

● מחשב ATARI ST

יצא לאור "שעשועי מחשבת-1"

"שעשועי מחשבת-1" – מציינת את

תחילתה של סדרה חדשה של

"מחשבת" ומיועדת לכל חובבי

המחשבים הביתיים המחפשים אתגרים

תכנותיים.

מה תמצא בחוברת הראשונה של

"שעשועי מחשבת-1"?

הצעות מפורטות לפיתוח 5 משחקי

מחשב המתאימים מאד לתכנות בשפת

BASIC.

החוברת אינה כוללת אף תוכנית – את

זאת אנו משאירים לך לפתח.

בנוסף לכך תוכל להשתתף בהגרלה

נושאת פרסים של "מחשבת-BUG".

(פרסים תמצא בחוברת עצמה).





באמצעות חוברת זו ואלו שיבואו בעקבותיה
תלמד לכתוב תוכניות למחשב בתחומים שונים
ומגוונים, למשל, ציור בעזרת מחשב, פיתוח משחקי
מחשב טלוויזיה, משחקים דיסקטיים, חישובים מתמטיים,
עיבוד נתונים ותוך כדי כך תלמד
את שפת ה-BASIC.

איזה רקע נדרש ממך?

- למעשה כמעט ולא נדרש ממך ידע מוקדם:
- אינך צריך להבין דבר וחצי דבר במחשבים.
 - אינך צריך לדעת מתמטיקה.
- וכל זאת, כיוון שביססנו את הלימוד על פיתוח
המושגים בגישה גרפית ומשחקית.